
**************************************************************************; 
* program:    Create combined consumption tax base                        ; 
* programmer: Rick Peterson                                               ; 
* project:    Washington Excise Tax Microsimulation Model                 ; 
* date:       March 27, 2002                                              ; 
*                                                                         ; 
* purpose:    Creates an excise tax base for each of the seven matched    ; 
*             SPS data sets                                               ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* libraries:   extaxmdl - location of excise tax model data sets          ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* incoming:    expn1-expn4 - Diary survey consumption data                ; 
*              mtab1-mtab4,Itab1-Itab4 - Interview survey consumption data; 
*              Taxable items list.xls - Excise tax bases by UCC codes     ; 
*-------------------------------------------------------------------------; 
* formats:    None                                                        ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* outgoing:   extaxmdl.taxbase1 - taxbase7  - One data set for each of    ; 
*             the seven SPS matched data set                              ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* reports:   None                                                         ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* changes:                                                                ; 
*                                                                         ; 
*-------------------------------------------------------------------------; 
* notes:     The program first combines consumption from the Diary and    ; 
*            Interview survey. Diary data is used for food, and for items ; 
*            not included in the Interview survey. Otherwise consumption  ; 
*            data from the Interview survey is used. The combined         ; 
*            consumption data is matched with the external file which     ; 
*            identifies UCCs associated with the tax bases for the excise ; 
*            taxes.  This file, called taxbase, contains all the taxable  ; 
*            consumption for each CU in the CEX quarterly data sets.      : 
*            This file is then separated into diary and interview         ; 
*            data sets by quarter.  Then the data is matched with the SPS ; 
*            identification numbers created in hot decking programs.      ; 
*            Data is summarized by SPS id number for each of the seven    ; 
*            sets of matched data sets.                                   ; 
**************************************************************************; 
*-------------------------------------------------------------------------; 
*Read in consumption data from Diary data sets; 
*-------------------------------------------------------------------------; 
%let y = 99; 
filename expn1 "c:\data\diary survey\expnd&y.1.txt"; 
filename expn2 "c:\data\diary survey\expnd&y.2.txt"; 
filename expn3 "c:\data\diary survey\expnd&y.3.txt"; 

filename expn4 "c:\data\diary survey\expnd&y.4.txt"; 
%macro loop; 
%do x = 1 %to 4; 
data expn&x; 
   infile expn&x lrecl=40; 
   input @1 newid 8. @35 ucc $6. @10 cost 12.5 ; 
   dataset = "D&x"; 
   run; 
%end; 
%mend Loop; 
%Loop; 
 
*-------------------------------------------------------------------------; 
*Combine the diary consumption data sets; 
*Keep only selected data from Diary Survey; 
*-------------------------------------------------------------------------; 
 
Data Diarycombined; 
set expn1 expn2 expn3 expn4; 
ucccode=input(ucc,$6.); 
if 10110 <=ucccode<=200536 then output; 
else if ucccode in (220400, 220410, 270905, 320380, 320430) then output; 
else if 320905 <= ucccode <= 330510 then output; 
else if 330610 <= ucccode <=340120 then output; 
else if ucccode in (340913, 450900, 470114, 480212, 490000, 490316, 
520541, 530903, 550210, 550310, 550410, 550900, 570902, 590900, 
610220, 610310, 610901, 610902, 610903, 620510, 620610, 620710, 
620911, 620913, 630220, 630900, 640110, 640120, 640220, 640310, 
640410, 660000, 680903) then output; 
run; 
 
*-------------------------------------------------------------------------; 
*Read in consumption data from Interview data sets; 
*-------------------------------------------------------------------------; 
 
filename mtab1 "c:\data\interview survey\mtabi&y.1x.txt"; 
filename mtab2 "c:\data\interview survey\mtabi&y.2.txt"; 
filename mtab3 "c:\data\interview survey\mtabi&y.3.txt"; 
filename mtab4 "c:\data\interview survey\mtabi&y.4.txt"; 
 
%macro loop2; 
%do x= 1 %to 4 %by 1; 
data mtab&x; 

infile mtab&x lrecl=35; 
input @1 newid 8.  

@9 ucc $6.  
@15 cost 12.4; 
dataset = "I&x"; 

proc sort; 
by newid; 
run; 
%end; 
%mend Loop2; 

%Loop2; 
 
filename itab1 "c:\data\interview survey\itabi&y.1x.txt"; 
filename itab2 "c:\data\interview survey\itabi&y.2.txt"; 
filename itab3 "c:\data\interview survey\itabi&y.3.txt"; 
filename itab4 "c:\data\interview survey\itabi&y.4.txt"; 
 
%macro loop3; 
%do x= 1 %to 4 %by 1; 
data itab&x; 

infile itab&x lrecl=35; 
input @1 newid 8.  

@15 ucc $6.  
@22 cost 12.4; 

        dataset = "I&x"; 
proc sort; 
by newid; 
run; 
%end; 
%mend Loop3; 
%Loop3; 
 
Data Interviewcombined; 
set mtab1 mtab2 mtab3 mtab4  
itab1 itab2 itab3 itab4; 
by newid; 
if ucc='710110' then cost =cost*4; 
run; 
 
*-------------------------------------------------------------------------; 
*Combine Diary and Interview data; 
*-------------------------------------------------------------------------; 
 
data extaxmdl.consumption; 
set diarycombined interviewcombined;  
run; 
 
proc sort data=extaxmdl.consumption; 
by ucc; 
run; 
 
*-------------------------------------------------------------------------; 
*Delete preliminary data sets; 
*-------------------------------------------------------------------------; 
proc datasets; 
delete expn1 expn2 expn3 expn4 mtab1 mtab2 mtab3 mtab4  
itab1 itab2 itab3 itab4 diarycombined interviewcombined; 
run; 

*-------------------------------------------------------------------------; 
*Import taxable items list from spreadsheet; 
*-------------------------------------------------------------------------; 
 
proc import datafile="c:\data\excise tax model\Taxable items list.xls" 
out=taxed_items replace; 
run; 
  
*Change formats of ucc, survey, and labels; 
 
data taxed_items; 
set taxed_items; 
attrib ucc length = $6; 
ucc=put(ucccode,6.); 
if length(left(ucc))=4 then ucc='00'||trim(left(ucc)); 
if length(left(ucc))=5 then ucc='0'||trim(left(ucc)); 
drop ucccode surveycode ucccodelabel; 
run; 
 
 
*-------------------------------------------------------------------------; 
*Merge consumption data set and taxable items list; 
*Calculate tax bases and taxes; 
*-------------------------------------------------------------------------; 
 
data extaxmdl.taxbase; 
merge extaxmdl.consumption taxed_items; 
by ucc; 
*Sales tax base; 

Salesbase=Baseadj*current*cost/(1+.084); 
Alt1base=Baseadj*Alt1*cost/(1+.084); 
Alt2base=Baseadj*cost/(1+.084); 

 
*Alcohol tax bases; 

*Liquor sales and liter in container; 
*Base calculated at average price of $10 per 750 ml bottle; 
if other =3 then liquorsalesbase_container = Baseadj*cost/(1+.205); 
else liquorsalesbase_container = 0; 
if other = 3 then liquorvolbase_container = (Baseadj*cost/10)*.750; 
else liquorvolbase_container = 0; 

 
*Liquor sales and liter by the drink; 
*Base calculated assuming 1.5 oz drinks; 

    *which equals an average of 16.8 drinks per 750 ml bottle; 
*at $3.25 per drink; 
if other = 6 then liquorsalesbase_drink = Baseadj*cost/(1+.084); 
else liquorsalesbase_drink = 0; 
if other = 6 then liquorvolbase_drink = ((Baseadj*cost/3.25)/16.8)*.750; 
else liquorvolbase_drink = 0; 

*Wine tax; 
*Wine sold in containers is based on average  
price of $5.00, including tax, for a 750 ml bottle; 
*Wine by the glass - assumed effective price of $20 per 750 ml bottle; 
if other = 2 then Wineliterbase = (Baseadj*cost/5)*.75; 
Else if other = 5 then Wineliterbase = (Baseadj*cost/20)*.75; 
Else Wineliterbase =0; 

 
*Beer Tax - rate per 31 gallons; 
*Note - a six pack of 12oz bottles is 72 oz. which is (72/128) gallons; 
*Assume $4 price per six pack for beer sold in containers; 
*Assume $8 price per six pack for beer sold by drink; 
if other = 1 then beerbase = (Baseadj*cost/4)*(72/128)/31; 
Else if other =4 then beerbase = (Baseadj*cost/8)*(72/128)/31; 
Else beerbase = 0; 
 

*Insurance Tax; 
if other = 7 then Insbase = Baseadj*cost/(1+.02); 
Else Insbase = 0;  

 
*Cigarette and other tobacco products Tax; 

*Assume cigarette pack price is $4; 
if other = 8 then cigbase = Baseadj*cost/4; 
else cigbase = 0; 

 
*Assume wholesale mark up is 50% on other tobacco products; 
if other = 9 then othertobbase = 0.842*(Baseadj*cost/1.5)/(1+.749); 
else othertobbase = 0; 

 
*Real Estate Excise Tax; 

if other = 10 then Reetbase = 10*Baseadj*cost; 
Else reetbase = 0; 

 
*Public Utility Tax; 

if other = 11 then elecbase=Baseadj*cost/(1+.03893); 
else elecbase = 0; 
if other = 12 then naturalgasbase = Baseadj*cost/(1+.03852); 
else naturalgasbase = 0; 
if other = 13 then waterseweragebase = 

Baseadj*cost/(1+(.05029+.03852)/2); 
else waterseweragebase = 0; 
if other = 14 then garbagebase = Baseadj*cost/(1+.036); 
else garbagebase = 0; 
if other = 17 then intercitybase = Baseadj*cost/(1+.01926); 
else intercitybase = 0; 
if other = 18 then intracitybase =  Baseadj*cost/(1+.00642); 
else intracitybase = 0;  

*Gas Tax; 
if other = 15 then gasbase = Baseadj*cost/1.4; 
else gasbase = 0 ; 

run; 
 
*-------------------------------------------------------------------------; 
*Separate data into quarterly Diary and Interview data sets; 
*Summarize tax bases by CEX id number; 
*-------------------------------------------------------------------------; 
 
%macro loop; 
%do x = 1 %to 4; 
 
data a; 
set extaxmdl.taxbase; 
where dataset = "D&x"; 
proc summary data=a; 
class newid; 
var Salesbase Alt1base Alt2base liquorsalesbase_container 
liquorvolbase_container  
liquorsalesbase_drink liquorvolbase_drink 
Wineliterbase beerbase Insbase cigbase othertobbase reetbase elecbase 
 naturalgasbase waterseweragebase garbagebase intercitybase intracitybase  
 gasbase; 
output out=extaxmdl.newidDiary&x sum=; 
proc sort data=extaxmdl.newidDiary&x;  
by newid; 
run; 
 
data a; 
set extaxmdl.taxbase; 
where dataset="I&x"; 
proc summary data=a; 
class newid; 
var Salesbase Alt1base Alt2base liquorsalesbase_container 
liquorvolbase_container  
liquorsalesbase_drink liquorvolbase_drink 
Wineliterbase beerbase Insbase cigbase othertobbase reetbase elecbase 
 naturalgasbase waterseweragebase garbagebase intercitybase intracitybase  
 gasbase; 
output out=extaxmdl.newidInterv&x sum=; 
proc sort data=extaxmdl.newidInterv&x;  
by newid; 
run; 
 
%end; 
%mend Loop; 
%Loop; 

*-------------------------------------------------------------------------; 
*Separate the seven matched SPS data sets into 4 X 2 X 7 data sets. One data  
set data for each matched group by quarter by survey type; 
*-------------------------------------------------------------------------; 
 
%macro Loop2; 
%do x= 1 %to 4; 
%do z = 1 %to 7; 
Data one; 
set mdimpute.dimpute&x; 
rename newid&z = newid; 
keep id newid&z; 
Proc sort data=one  out=D&x&z; 
by newid; 
run; 
Data one; 
set miimpute.iimpute&x; 
rename newid&z = newid; 
keep id newid&z; 
Proc sort data=one  out=I&x&z; 
by newid; 
run; 
%end; 
%end; 
%mend Loop2; 
%Loop2; 
 
*-------------------------------------------------------------------------; 
*Create 2 (D,I) by 4 (qtrs) by 7 (number of groups) data sets from  
consumption data with each WAPOP household associated with summary  
expenditure data for the NEWID attached to the WAPOP household; 
*-------------------------------------------------------------------------; 
 
%macro Loop3; 
%do x = 1 %to 4; 
%do z = 1 %to 7; 
Proc SQL noprint; 
create table work.DA&x&z as 
select * 
  from  D&x&z , extaxmdl.newidDiary&x  
  where D&x&z..newid = newidDiary&x..newid ; 
  quit; 
Proc sort data=DA&x&z; 
by id; 
Proc SQL noprint; 
create table work.IA&x&z as 
select * 
  from  I&x&z , extaxmdl.newidinterv&x  
  where I&x&z..newid = newidinterv&x..newid ; 

  quit; 
Proc sort data=IA&x&z; 
by id; 
run; 
%end; 
%end; 
%mend Loop3; 
%Loop3; 
 
*-------------------------------------------------------------------------; 
*Combine Diary and Interview data for each SPS match group; 
*First combine the 4 qtrs of consumtion data by Diary and Interview  
 and sum by SPS id; 
*Second, merge in information on the number of qtrs sucessfully matched 
 by Diary and Interview and adjust the data accordingly; 
*Combine Diary and Interview data and sum by SPS id; 
*-------------------------------------------------------------------------; 
 
%macro loop4; 
%do x=1 %to 7; 
data combine_qtrs; 
set DA1&x DA2&x DA3&x DA4&x; 
by id; 
 
proc summary data=combine_qtrs; 
by id; 
var Salesbase Alt1base Alt2base liquorsalesbase_container 
liquorvolbase_container  
liquorsalesbase_drink liquorvolbase_drink 
Wineliterbase beerbase Insbase cigbase othertobbase reetbase elecbase 
 naturalgasbase waterseweragebase garbagebase intercitybase intracitybase  
 gasbase; 
output out = sumcombined sum=; 
 
data dcombined&x; 
merge sumcombined mdimpute.dqtrsmatched; 
by id; 
  costadj = 52/dqtrs; 
  Salesbase = costadj*Salesbase; 
  Alt1base  = costadj*Alt1base; 
  Alt2base  = costadj*Alt2base; 
  liquorsalesbase_container  = costadj*liquorsalesbase_container; 
  liquorvolbase_container  = costadj*liquorvolbase_container; 
  liquorsalesbase_drink  = costadj*liquorsalesbase_drink; 
  liquorvolbase_drink = costadj*liquorvolbase_drink; 
  Wineliterbase  = costadj*Wineliterbase; 
  beerbase = costadj*beerbase; 
  Insbase  = costadj*Insbase; 
  cigbase  = costadj*cigbase;  

  othertobbase  = costadj*othertobbase; 
  reetbase  = costadj*reetbase;   
  elecbase = costadj*elecbase;  
  naturalgasbase  = costadj*naturalgasbase; 
  waterseweragebase  = costadj*waterseweragebase; 
  garbagebase  = costadj*garbagebase; 
  intercitybase  = costadj*intercitybase; 
  intracitybase  = costadj*intracitybase; 
  gasbase = costadj*gasbase; 
 
*End of Diary processing and begining of Interview processing; 
 
data combine_qtrs; 
set IA1&x IA2&x IA3&x IA4&x ; 
by id; 
 
proc summary data=combine_qtrs; 
by id; 
var Salesbase Alt1base Alt2base liquorsalesbase_container 
liquorvolbase_container  
liquorsalesbase_drink liquorvolbase_drink 
Wineliterbase beerbase Insbase cigbase othertobbase reetbase elecbase 
 naturalgasbase waterseweragebase garbagebase intercitybase intracitybase  
 gasbase; 
output out = sumcombined sum=; 
 
data Icombined&X; 
merge sumcombined mIimpute.iqtrsmatched; 
by id; 
costadj = 4/dqtrs; 
  Salesbase = costadj*Salesbase; 
  Alt1base  = costadj*Alt1base; 
  Alt2base  = costadj*Alt2base; 
  liquorsalesbase_container  = costadj*liquorsalesbase_container; 
  liquorvolbase_container  = costadj*liquorvolbase_container; 
  liquorsalesbase_drink  = costadj*liquorsalesbase_drink; 
  liquorvolbase_drink = costadj*liquorvolbase_drink; 
  Wineliterbase  = costadj*Wineliterbase; 
  beerbase = costadj*beerbase; 
  Insbase  = costadj*Insbase; 
  cigbase  = costadj*cigbase;  
  othertobbase  = costadj*othertobbase; 
  reetbase  = costadj*reetbase;   
  elecbase = costadj*elecbase;  
  naturalgasbase  = costadj*naturalgasbase; 
  waterseweragebase  = costadj*waterseweragebase; 
  garbagebase  = costadj*garbagebase; 
  intercitybase  = costadj*intercitybase; 
  intracitybase  = costadj*intracitybase; 

  gasbase = costadj*gasbase; 
 
*End of Interview processing; 
*Combine adjusted Diary and Interview data and sum by id; 
 
data combined&x; 
set dcombined&x icombined&x; 
by id; 
 
Proc summary data=combined&x; 
by id; 
var Salesbase Alt1base Alt2base liquorsalesbase_container 
liquorvolbase_container  
liquorsalesbase_drink liquorvolbase_drink 
Wineliterbase beerbase Insbase cigbase othertobbase reetbase elecbase 
 naturalgasbase waterseweragebase garbagebase intercitybase intracitybase  
 gasbase; 
output out=extaxmdl.taxbase&x sum=; 
run; 
%end; 
%mend loop4; 
%loop4; 
 
 




